Oscillatori.

Consideriamo il circuito rappresentato in figura, costituito da una maglia con quattro componenti in serie: un generatore (indipendente) di tensione, un induttore, un condensatore e un bipolo resistivo (la cui caratteristica sarà poi precisata).



I componenti reattivi (induttivi o capacitivi) sono in grado di immagazzinare energia (magnetica o elettrica, rispettivamente); le loro equazioni descrittive sono:

$$v_L(t) = L \cdot di(t)/dt \;\; ; \quad i(t) = C \cdot dv_C(t)/dt$$

(la corrente che li attraversa qui è la stessa). Inoltre, per la legge di Kirchhoff delle tensioni, in ogni istante di tempo si ha:

$$V_g = v_L + v_C + f(i)$$

Consideriamo dapprima il caso in cui $f(i) = R \cdot i(t)$ (bipolo puramente resistivo, che dissipa energia).
Se, ad esempio, la tensione V_g erogata dal generatore è costante nel tempo, a regime raggiunto nella maglia non circola più corrente ($i = 0$) e tutta la tensione cade ai capi del condensatore ($v_C = V_g$). Comunque sia, se nell'istante per noi iniziale ($t = 0$) chiudiamo il generatore (vale a dire: lo sostituiamo con un *cortocircuito*, un "filo ideale" privo di resistenza), allora $V_g \equiv 0$ per $t \geq 0$ e i componenti reattivi iniziano a scaricarsi (se non sono già scarichi); quindi, per $t \geq 0$:

$$0 = L \cdot di/dt + v_C + R \cdot i$$

per cui si ha il sistema *lineare* autonomo:

$$\begin{cases} \dfrac{di}{dt} = -\dfrac{R}{L}\, i - \dfrac{1}{L}\, v_C \\[2ex] \dfrac{dv_C}{dt} = \dfrac{1}{C}\, i \end{cases}$$

Le variabili di stato sono la corrente che attraversa l'induttore (nonché, nel nostro caso, gli altri componenti dell'unica maglia del circuito) e la tensione ai capi del condensatore; chiamandole rispettivamente x e y (funzioni del tempo), e assumendo unitarie l'induttanza (L = 1 H = 1 Ω·s) e la capacità (C = 1 F = 1 s/Ω), il sistema si riscrive così:

$$\begin{cases} \dfrac{\mathrm{d}x}{\mathrm{d}t} = -\mathsf{R}\,x - y \\[2ex] \dfrac{\mathrm{d}y}{\mathrm{d}t} = x \end{cases}$$

Le isocline nulle sono la retta di equazione $y = -\mathsf{R}\,x$ e l'asse y, che si intersecano nell'origine: dunque l'origine è l'unico punto di equilibrio, qualunque sia R.
La retta $y = -\mathsf{R}\,x$ si ottiene ribaltando la caratteristica del resistore lineare nel piano (i, v) rispetto all'asse verticale (v): tale caratteristica è infatti l'equazione $v = \mathsf{R} \cdot i$, che nel piano (i, v) è una retta passante per l'origine e tutta contenuta nel primo e terzo quadrante: ciò che è peculiare di un componente *passivo*…
Vediamo l'evoluzione del sistema, sia sul piano delle fasi sia nel tempo, a partire da diverse condizioni iniziali − lo stato iniziale è costituito dalla corrente e dalla tensione ai capi del condensatore per t = 0, e il circuito è già a riposo se e soltanto se esse sono entrambe nulle − per diversi valori (positivi) della resistenza R (che è il nostro parametro) e poi nel caso particolare (ideale!) in cui R = 0. (Si immagini sempre di tracciare le isocline nulle sul piano dove è raffigurato il campo vettoriale.)
Iniziamo con R = 1 Ω.

```
restart:
with(DEtools):
R := 1:
RLC := diff(x(t), t) = -R*x(t)-y(t), diff(y(t), t) = x(t):
vars := x(t), y(t):
DEplot({RLC}, {vars}, t = 0..10, x = -3..3, y = -3..3, title =
  `Serie RLC: x = corrente, y = tensione su C`, colour = magnitude);
```



```
DEplot({RLC}, {vars}, t = 0..10, x = -3..3, y = -3..3, [[x(0)=3, y(0)=0],
 [x(0)=2, y(0)=2], [x(0)=0, y(0)=3], [x(0)=-2, y(0)=2], [x(0)=-3, y(0)=0]],
 animate = true, numframes = 250, arrows = none, thickness = 1,
 linecolour = [red, blue, red, blue, red]);
```




```
with(plots):
sol := dsolve({x(0)=0, y(0)=3, RLC}, {vars},
              type = numeric, output = listprocedure):
corrente := subs(sol, x(t)):
tensione := subs(sol, y(t)):
plot([corrente, tensione], 0 .. 15, colour = [cyan, magenta],
     title = `cyan = corrente, x(0) = 0; magenta = tensione, y(0) = 3`);
```



Ripetendo la stessa procedura con `R:=2:` e con `R:=3:`, otteniamo le figure appresso riportate, nella colonna di sinistra e in quella di destra, rispettivamente.



Per ogni $R > 0$, tutte le soluzioni del sistema tendono a zero: fisicamente, ciò è dovuto all'effetto dissipativo del resistore.
In particolare, si verifica che, se $0 < R < 2\ \Omega$ (si provi anche, ad esempio, a fissare `R:=0.5`), allora le traiettorie sono delle spirali che convergono al punto di equilibrio, che è un *fuoco* stabile (nel tempo, le variabili di stato si smorzano e vanno a zero oscillando); quando invece $R \geq 2\ \Omega$ non si hanno più oscillazioni: il punto di equilibrio è un *nodo* stabile.
Man mano che R si avvicina a zero, le oscillazioni si smorzano sì, ma sempre più lentamente, finché quando $R = 0$ il sistema diventa un *oscillatore armonico* ideale, che conserva l'energia: in tal caso, le isocline nulle sono i due assi e le traiettorie costituiscono una famiglia continua di orbite perfettamente circolari con centro nell'origine (che è appunto un *centro*, neutralmente stabile). Si ripeta dunque la procedura ponendo `R:=0:`, e si osservi, in particolare, che la tensione ai capi del condensatore è in ritardo di un quarto di periodo rispetto alla corrente.